Gaussian Quadrature

This configuration is widely used. In 2D problem, integration is done using values of “f” at 4 quad. points.



Gaussian Quadrature

This configuration is widely used. In 2D problem, integration is done using values of “f” at 4 quad. points.



$$I=\int_{-1}^{+1}\int_{-1}^{+1} f(\xi,\eta)\,d\xi d\eta=\sum_{i=1}^{m}\sum_{j=1}^{n}\left[W_i\cdot W_j\cdot f(\xi_i,\eta_j)\right]$$

$$=1.0\times1.0\times f(-0.57735,-0.57735)+1.0\times1.0\times f(-0.57735,+0.57735)$$

$$+1.0\times1.0\times f(+0.57735,+0.57735)+1.0\times1.0\times f(+0.57735,-0.57735)$$

Homework

- Develop a program and calculate area of the following quadrilateral using Gaussian Quadrature.



1：(1.0,　1.0)
2：(4.0,　2.0)
3：(3.0,　5.0)
4：(2.0,　4.0)

$$I = \int_V dV$$

Tips

- Calculate Jacobian
- Apply Gaussian Quadrature (n=2)

$$I = \int_{-1}^{+1}\int_{-1}^{+1} f(\xi,\eta)\, d\xi d\eta = \sum_{i=1}^{m}\sum_{j=1}^{n}\left[W_i \cdot W_j \cdot f(\xi_i,\eta_j)\right]$$

```
implicit REAL*8 (A-H,O-Z)
real*8 W(2)
real*8 POI(2)

W(1)= 1.0d0
W(2)= 1.0d0
POI(1)= -0.5773502692d0
POI(2)= +0.5773502692d0

SUM= 0.d0
do jp= 1, 2
do ip= 1, 2
  FC = F(POI(ip),POI(jp))
  SUM= SUM + W (ip)*W (jp)*FC
enddo
enddo
```



Results

1：(1.0,　1.0)
2：(4.0,　2.0)
3：(3.0,　5.0)
4：(2.0,　4.0)



$$3\times 4-\frac{1}{2}\left(3+3+2+4\right)\times 1$$

$$=12-\frac{12}{2}=6$$

What to do ?

- Final Target:

$$I = \int_V dV = \iint dxdy = \int_{-1}^{+1}\int_{-1}^{+1} \det[J]\, d\xi d\eta$$

- By Definition:

$$I = \int_{-1}^{+1}\int_{-1}^{+1} f(\xi,\eta)\, d\xi d\eta = \sum_{i=1}^{m}\sum_{j=1}^{n}\left[W_i \cdot W_j \cdot f(\xi_i,\eta_j)\right]$$

- Therefore, det [*J*]= *f*
- Value of det [*J*] at quad. points should be calculated !

$$\det\left[J(\xi_i,\eta_j)\right]$$

Initialization (1/4)

```
      implicit REAL*8 (A-H,O-Z)

      real*8 X(4), Y(4)
      real*8 W(2), POS(2)
      real*8 SHAPE(2,2,4)
      real*8 PNQ(2,4), PNE(2,4), DETJ(2,2)

!C
!C-- POINT data
      X(1)= 1.0
      Y(1)= 1.0

      X(2)= 4.0
      Y(2)= 2.0

      X(3)= 3.0
      Y(3)= 5.0

      X(4)= 2.0
      Y(4)= 4.0

!C
!C-- Quadrature points & weighting coef.
      W(1)= +1.0000000000d+00
      W(2)= +1.0000000000d+00

      POS(1)= -0.5773502692d+00
      POS(2)= +0.5773502692d+00
```

Node Coord.

1: (1.0, 1.0)
2: (4.0, 2.0)
3: (3.0, 5.0)
4: (2.0, 4.0)



Initialization (1/4)

```
      implicit REAL*8 (A-H,O-Z)

      real*8 X(4), Y(4)
      real*8 W(2), POS(2)
      real*8 SHAPE(2,2,4)
      real*8 PNQ(2,4), PNE(2,4), DETJ(2,2)

!C
!C-- POINT data
      X(1)= 1.0
      Y(1)= 1.0

      X(2)= 4.0
      Y(2)= 2.0

      X(3)= 3.0
      Y(3)= 5.0

      X(4)= 2.0
      Y(4)= 4.0

!C
!C-- Quadrature points & weighting coef.
      W(1)= +1.0000000000d+00
      W(2)= +1.0000000000d+00

      POS(1)= -0.5773502692d+00
      POS(2)= +0.5773502692d+00
```

POS: Quad. Points

W: Weighting Factors.



Shape Fn., Derivatives at Quad. Points (2/4)

```
!C
!C-- SHAPE functions
      O4th= 0.25d0

      do jp= 1, 2
      do ip= 1, 2
        QP1= 1.d0 + POS(ip)
        QM1= 1.d0 - POS(ip)
        EP1= 1.d0 + POS(jp)
        EM1= 1.d0 - POS(jp)

        SHAPE(ip,jp,1)= O4th * QM1 * EM1
        SHAPE(ip,jp,2)= O4th * QP1 * EM1
        SHAPE(ip,jp,3)= O4th * QP1 * EP1
        SHAPE(ip,jp,4)= O4th * QM1 * EP1

        PNQ(jp,1)= - O4th * EM1
        PNQ(jp,2)= + O4th * EM1
        PNQ(jp,3)= + O4th * EP1
        PNQ(jp,4)= - O4th * EP1

        PNE(ip,1)= - O4th * QM1
        PNE(ip,2)= - O4th * QP1
        PNE(ip,3)= + O4th * QP1
        PNE(ip,4)= + O4th * QM1
      enddo
      enddo
```

$$QP1(i) = (1 + \xi_i), \quad QM1(i) = (1 - \xi_i)$$

$$EP1(j) = (1 + \eta_j), \quad EM1(j) = (1 - \eta_i)$$

Shape Fn., Derivatives at Quad. Points (2/4)

```
!C
!C-- SHAPE functions
      O4th= 0.25d0

      do jp= 1, 2
      do ip= 1, 2
        QP1= 1.d0 + POS(ip)
        QM1= 1.d0 - POS(ip)
        EP1= 1.d0 + POS(jp)
        EM1= 1.d0 - POS(jp)

        SHAPE(ip,jp,1)= O4th * QM1 * EM1
        SHAPE(ip,jp,2)= O4th * QP1 * EM1
        SHAPE(ip,jp,3)= O4th * QP1 * EP1
        SHAPE(ip,jp,4)= O4th * QM1 * EP1

        PNQ(jp,1)= - O4th * EM1
        PNQ(jp,2)= + O4th * EM1
        PNQ(jp,3)= + O4th * EP1
        PNQ(jp,4)= - O4th * EP1

        PNE(ip,1)= - O4th * QM1
        PNE(ip,2)= - O4th * QP1
        PNE(ip,3)= + O4th * QP1
        PNE(ip,4)= + O4th * QM1
      enddo
      enddo
```

SHAPE: Values of Shape Fn's.@(ξ_i, η_j)

$$N_1(\xi_i,\eta_j) = \frac{1}{4}(1-\xi_i)(1-\eta_j)$$

$$N_2(\xi_i,\eta_j) = \frac{1}{4}(1+\xi_i)(1-\eta_j)$$

$$N_3(\xi_i,\eta_j) = \frac{1}{4}(1+\xi_i)(1+\eta_j)$$

$$N_4(\xi_i,\eta_j) = \frac{1}{4}(1-\xi_i)(1+\eta_j)$$



Shape Fn., Derivatives at Quad. Points (2/4)

```
!C
!C-- SHAPE functions
      O4th= 0.25d0

      do jp= 1, 2
      do ip= 1, 2
        QP1= 1.d0 + POS(ip)
        QM1= 1.d0 - POS(ip)
        EP1= 1.d0 + POS(jp)
        EM1= 1.d0 - POS(jp)

        SHAPE(ip,jp,1)= O4th * QM1 * EM1
        SHAPE(ip,jp,2)= O4th * QP1 * EM1
        SHAPE(ip,jp,3)= O4th * QP1 * EP1
        SHAPE(ip,jp,4)= O4th * QM1 * EP1

        PNQ(jp,1)= - O4th * EM1
        PNQ(jp,2)= + O4th * EM1
        PNQ(jp,3)= + O4th * EP1
        PNQ(jp,4)= - O4th * EP1

        PNE(ip,1)= - O4th * QM1
        PNE(ip,2)= - O4th * QP1
        PNE(ip,3)= + O4th * QP1
        PNE(ip,4)= + O4th * QM1
      enddo
      enddo
```

$$PNQ(j,k) = \frac{\partial N_k}{\partial \xi}(\xi = \xi_i, \eta = \eta_j)$$

$$PNE(j,k) = \frac{\partial N_k}{\partial \eta}(\xi = \xi_i, \eta = \eta_j)$$

$$\frac{\partial N_1}{\partial \xi}(\xi_i,\eta_j) = -\frac{1}{4}(1-\eta_j) \qquad \frac{\partial N_1}{\partial \eta}(\xi_i,\eta_j) = -\frac{1}{4}(1-\xi_i)$$

$$\frac{\partial N_2}{\partial \xi}(\xi_i,\eta_j) = +\frac{1}{4}(1-\eta_j) \qquad \frac{\partial N_2}{\partial \eta}(\xi_i,\eta_j) = -\frac{1}{4}(1+\xi_i)$$

$$\frac{\partial N_3}{\partial \xi}(\xi_i,\eta_j) = +\frac{1}{4}(1+\eta_j) \qquad \frac{\partial N_3}{\partial \eta}(\xi_i,\eta_j) = +\frac{1}{4}(1+\xi_i)$$

$$\frac{\partial N_4}{\partial \xi}(\xi_i,\eta_j) = -\frac{1}{4}(1+\eta_j) \qquad \frac{\partial N_4}{\partial \eta}(\xi_i,\eta_j) = +\frac{1}{4}(1-\xi_i)$$

1st order Derivatives at (ξ_i, η_j)

Jacobian at Quad. Points (3/4)

```
!C
!C +-----------------+
!C | JACOBIAN matrix |
!C +-----------------+
!C===
      do jp= 1, 2
      do ip= 1, 2
        dXdQ = PNQ(jp,1)*X(1) + PNQ(jp,2)*X(2) +
     &         PNQ(jp,3)*X(3) + PNQ(jp,4)*X(4)
        dYdQ = PNQ(jp,1)*Y(1) + PNQ(jp,2)*Y(2) +
     &         PNQ(jp,3)*Y(3) + PNQ(jp,4)*Y(4)
        dXdE = PNE(ip,1)*X(1) + PNE(ip,2)*X(2) +
     &         PNE(ip,3)*X(3) + PNE(ip,4)*X(4)
        dYdE = PNE(ip,1)*Y(1) + PNE(ip,2)*Y(2) +
     &         PNE(ip,3)*Y(3) + PNE(ip,4)*Y(4)
        DETJ(ip,jp)= dXdQ*dYdE - dXdE*dYdQ
      enddo
      enddo
!C===
```

$$dXdQ = \frac{\partial x}{\partial \xi} \quad dYdQ = \frac{\partial y}{\partial \xi}$$

$$dXdE = \frac{\partial x}{\partial \eta} \quad dYdE = \frac{\partial y}{\partial \eta}$$

$$DETJ(i,j) = \det\left[J\left(\xi_i, \eta_j\right)\right]$$

$$J_{11} = \frac{\partial x}{\partial \xi} = \frac{\partial}{\partial \xi}\left(\sum_{i=1}^{4} N_i x_i\right) = \sum_{i=1}^{4} \frac{\partial N_i}{\partial \xi} x_i, \quad J_{12} = \frac{\partial y}{\partial \xi} = \frac{\partial}{\partial \xi}\left(\sum_{i=1}^{4} N_i y_i\right) = \sum_{i=1}^{4} \frac{\partial N_i}{\partial \xi} y_i,$$

$$J_{21} = \frac{\partial x}{\partial \eta} = \frac{\partial}{\partial \eta}\left(\sum_{i=1}^{4} N_i x_i\right) = \sum_{i=1}^{4} \frac{\partial N_i}{\partial \eta} x_i, \quad J_{22} = \frac{\partial y}{\partial \eta} = \frac{\partial}{\partial \eta}\left(\sum_{i=1}^{4} N_i y_i\right) = \sum_{i=1}^{4} \frac{\partial N_i}{\partial \eta} y_i$$

Numerical Integration (4/4)

```
!C
!C +-------+
!C | AREA |
!C +-------+
!C===
      AREA= 0.d0
      do jp= 1, 2
      do ip= 1, 2
        AREA= AREA + dabs(DETJ(ip,jp)) * W(ip) * W(jp)
      enddo
      enddo

!C
!C-- ANALYTICAL SOLUTION
      XA2= X(2) - X(1)
      YA2= Y(2) - Y(1)
      XA3= X(3) - X(1)
      YA3= Y(3) - Y(1)
      XA4= X(4) - X(1)
      YA4= Y(4) - Y(1)

      AREAa= 0.50d0 * (dabs(XA2*YA3-YA2*XA3) +dabs(XA3*YA4-YA3*XA4))
!C===

      write (*,'(a,1pe16.6)') 'Gaussian quadrature', AREA
      write (*,'(a,1pe16.6)') 'analytical sol.    ', AREAa

      stop
      end
```

$$I = \int_{-1}^{+1} \int_{-1}^{+1} f(\xi,\eta)\, d\xi d\eta = \sum_{i=1}^{m} \sum_{j=1}^{n} \left[W_i \cdot W_j \cdot f(\xi_i,\eta_j) \right]$$